```
Import JJIO
Class Complex
-- Name A Nonomo Us
-- Does Complex numbers

Box re ofType real is private -- Real part
Box im ofType real is private -- Imaginary

Constructor Complex (r,i) is public
   Slot r ofType real
   Slot i ofType real
-- Does construct, create, initialize
   Set re = r
   Set im = i
EndConstructor Complex


Function plus (that) ofClass Complex is public
   Slot that  ofClass Complex
   Box result ofClass Complex
-- Does add complex numbers
   New result ofClass Complex with
     ( re + that.re, im + that.im )
EndFunction plus


Routine add (x, y) is public
   Slot x ofClass Complex
   Slot y ofClass Complex
-- Does add two complex numbers
   Set re = x.re + y.re
   Set im = x.im + y.im
EndRoutine add


Function times (it) ofClass Complex is public
   Slot it    ofClass Complex
   Box result ofClass Complex
   Box realPart ofType real
   Box imagPart ofType real
-- Does multiply complex numbers
   Set realPart = re * it.re - im * it.im
   Set imagPart = re * it.im + im * it.re
   New result ofClass Complex with
     ( realPart, imagPart )
EndFunction times

Function magnitude (none) ofType real is public
   Box result ofType real
-- Does return magnitude of a complex number
   Set result = Sqrt ( re * re + im * im)
EndFunction magnitude

Routine show (none) is public
-- Does print out a complex number
   Output re
   Output " + i"
   Output im
   Outputln " " -- gap
EndRoutine show

Function conjugate (none) ofClass Complex is public
   Box result ofClass Complex
-- Does return complex conjugate
   New result ofClass Complex with (re, 0.0 - im)
EndFunction conjugate

Routine testComplex (none) is private
   Box A ofClass Complex
   Box B ofClass Complex
   Box C ofClass Complex
-- Does test the Complex class
Start
   New A ofClass Complex with (1.0, 2.0)
   New B ofClass Complex with (3.0, 4.0)
   New C ofClass Complex with (0.0, 0.0)
   Set C = A.plus(B)
   Call C.show
   Set C = C.plus(A)
   Call C.show
   Call C.add with (A,B)
   Call C.show
   Set  C = A.times(B)
   Call C.show
EndRoutine testComplex

EndClass Complex

-- a run
-- 4.0 + i6.0
-- 5.0 + i8.0
-- 4.0 + i6.0
-- -5.0 + i10.0
```